# $Id$
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=cdparanoia
pkgver=10.2
pkgrel=5
pkgdesc="Compact Disc Digital Audio extraction tool"
arch=('i686' 'x86_64')
url="http://www.xiph.org/paranoia/"
license=('GPL')
depends=('glibc')
options=('!makeflags' '!staticlibs')
source=(http://downloads.xiph.org/releases/cdparanoia/cdparanoia-III-$pkgver.src.tgz
        gcc.patch)
sha1sums=('1901e20d3a370ca6afa4c76a9ef30d3f03044320'
          '67faaf0e3eed2e3de6f66d8168604bc6c7295921')

prepare() {
  cd "${srcdir}/${pkgname}-III-${pkgver}"
  patch -p0 -i ${srcdir}/gcc.patch
}

build() {
  cd "${srcdir}/${pkgname}-III-${pkgver}"
  ./configure --prefix=/usr --mandir=/usr/share/man
  make
}

package() {
  cd "${srcdir}/${pkgname}-III-${pkgver}"
  make prefix="${pkgdir}/usr" MANDIR="${pkgdir}/usr/share/man" install
}
